Mechanics - Morin 9.14 - Tennis Racket Theorem 11-05-18

N. T. Gladd

Initialization: Be sure the file *NTGUtilityFunctions.m* is in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Background

This is the fifth in a series of Mathematica notebooks on classical mechanics. This series was motivated by a close reading and problem solving project I undertook in 2014. The focus of my attention was the text ***Introduction to Classical Mechanics with Problems and Solutions***, by David Morin. This is a good book from which to learn and has a great collection of problems. I purchased it and recommend that those with interests in this topic acquire it for their library. I do note that an earlier version can be found on the web. This year, when I returned to this project, I decided to focus on generating Mathematica notebooks on material covered in *Chapter 9 Angular Momentum, Part II (General $\vec{L}$),* which deals the 3-D rigid body dynamics. This topic is notorious difficult/confusing for students and I felt I just skimmed by as a graduate student. I return in retirement after all these years to pay my dues and really understand how to solve problems is this area.

Although Morin’s Chapter 9 guides these notebooks, I made frequent use of other sources such as textbooks available in libraries or on the web. I also found lots of video lectures available on YouTube. Confused about some physics topics? Google it and you’ll be amazed what you find. Some relevant texts are

Classical Mechanics, Hebert Goldstein (my original text at University, late 60s). Newer versions exist.
Mechanics: Volume1 A Course in Theoretical Physics, L. D. Landau and E. M. Lifshitz.
Classical Mechanics, John. R. Taylor
Classical Mechanics of Particles and Systems, Stephen T. Thornton, Jerry B. Marion
Analytical Mechanics, G. R. Fowles, G. L. Cassiday

Analytical Mechanics, Louis N. Hand, Janet D. Finch

I find Mathematica useful for this topic. It facilitates calculations, provides a vehicle for creating instructive visualizations and allows one to quickly generate numerical solutions. Mathematica is a favorite tool of mine but I think it is crucially important to also work with pen and paper. Our brains are closely linked to our hands and one thinks differently with a pen in hand than when sitting before a computer screen. For serious thoughts on this, read *The Craftsman*, by Richard Sennett.

Tennis Racket Theorem

A rotating rigid body with three distinct moments of inertia $\mathcal{I}_1 > \mathcal{I}_2 > \mathcal{I}_3$ will be stable to rotation about the x_1 and x_3 principal axes, but unstable to rotation about the x_2 axis. I demonstrate this claim below.

The Euler equations are

```
template = τa == ℐa D[ωa[t], t] + (ℐc - ℐb) ωb[t] ωc[t];
w[1] = {template /. {a → 1, b → 2, c → 3},
   template /. {a → 2, b → 3, c → 1},
   template /. {a → 3, b → 1, c → 2}};
w[1] // MatrixForm
```

$$\begin{pmatrix} \tau_1 == (-\mathcal{I}_2 + \mathcal{I}_3)\,\omega_2[t]\,\omega_3[t] + \mathcal{I}_1\,\omega_1'[t] \\ \tau_2 == (\mathcal{I}_1 - \mathcal{I}_3)\,\omega_1[t]\,\omega_3[t] + \mathcal{I}_2\,\omega_2'[t] \\ \tau_3 == (-\mathcal{I}_1 + \mathcal{I}_2)\,\omega_1[t]\,\omega_2[t] + \mathcal{I}_3\,\omega_3'[t] \end{pmatrix}$$

```
w[2] = Solve[w[1], {ω1'[t], ω2'[t], ω3'[t]}][[1]]
```

$$\Big\{\omega_1'[t] \to \frac{1}{\mathcal{I}_1}\left(\tau_1 + \mathcal{I}_2\,\omega_2[t]\,\omega_3[t] - \mathcal{I}_3\,\omega_2[t]\,\omega_3[t]\right),$$
$$\omega_2'[t] \to \frac{1}{\mathcal{I}_2}\left(\tau_2 - \mathcal{I}_1\,\omega_1[t]\,\omega_3[t] + \mathcal{I}_3\,\omega_1[t]\,\omega_3[t]\right),$$
$$\omega_3'[t] \to \frac{1}{\mathcal{I}_3}\left(\tau_3 + \mathcal{I}_1\,\omega_1[t]\,\omega_2[t] - \mathcal{I}_2\,\omega_1[t]\,\omega_2[t]\right)\Big\}$$

Focus is on understanding whether the initial motion is stable or unstable. Perturb the angular velocities

```
perturbationRules =
 {ω1[t] → ω10[t] + ϵ δω1[t], ω2[t] → ω20[t] + ϵ δω2[t], ω3[t] → ω30[t] + ϵ δω3[t]}
```

$$\{\omega_1[t] \to \epsilon\,\delta\omega_1[t] + \omega_{10}[t],\ \omega_2[t] \to \epsilon\,\delta\omega_2[t] + \omega_{20}[t],\ \omega_3[t] \to \epsilon\,\delta\omega_3[t] + \omega_{30}[t]\}$$

```
w[3] = w[2] /. perturbationRules /. (D[#, t] & /@ perturbationRules)
```

$$\left\{\epsilon\,\delta\omega_1'[t]+\omega_{10}'[t]\to\frac{1}{I_1}\left(\tau_1+I_2\left(\epsilon\,\delta\omega_2[t]+\omega_{20}[t]\right)\left(\epsilon\,\delta\omega_3[t]+\omega_{30}[t]\right)-I_3\left(\epsilon\,\delta\omega_2[t]+\omega_{20}[t]\right)\left(\epsilon\,\delta\omega_3[t]+\omega_{30}[t]\right)\right),\right.$$
$$\epsilon\,\delta\omega_2'[t]+\omega_{20}'[t]\to\frac{1}{I_2}\left(\tau_2-I_1\left(\epsilon\,\delta\omega_1[t]+\omega_{10}[t]\right)\left(\epsilon\,\delta\omega_3[t]+\omega_{30}[t]\right)+I_3\left(\epsilon\,\delta\omega_1[t]+\omega_{10}[t]\right)\left(\epsilon\,\delta\omega_3[t]+\omega_{30}[t]\right)\right),$$
$$\left.\epsilon\,\delta\omega_3'[t]+\omega_{30}'[t]\to\frac{1}{I_3}\left(\tau_3+I_1\left(\epsilon\,\delta\omega_1[t]+\omega_{10}[t]\right)\left(\epsilon\,\delta\omega_2[t]+\omega_{20}[t]\right)-I_2\left(\epsilon\,\delta\omega_1[t]+\omega_{10}[t]\right)\left(\epsilon\,\delta\omega_2[t]+\omega_{20}[t]\right)\right)\right\}$$

Extract the linear terms

```
F[arg_Rule] := Coefficient[#, ϵ] & /@ arg;
w[4] = F /@ w[3]
```

$$\left\{\delta\omega_1'[t]\to\frac{1}{I_1}\left(I_2\,\delta\omega_3[t]\,\omega_{20}[t]-I_3\,\delta\omega_3[t]\,\omega_{20}[t]+I_2\,\delta\omega_2[t]\,\omega_{30}[t]-I_3\,\delta\omega_2[t]\,\omega_{30}[t]\right),\right.$$
$$\delta\omega_2'[t]\to\frac{1}{I_2}\left(-I_1\,\delta\omega_3[t]\,\omega_{10}[t]+I_3\,\delta\omega_3[t]\,\omega_{10}[t]-I_1\,\delta\omega_1[t]\,\omega_{30}[t]+I_3\,\delta\omega_1[t]\,\omega_{30}[t]\right),$$
$$\left.\delta\omega_3'[t]\to\frac{1}{I_3}\left(I_1\,\delta\omega_2[t]\,\omega_{10}[t]-I_2\,\delta\omega_2[t]\,\omega_{10}[t]+I_1\,\delta\omega_1[t]\,\omega_{20}[t]-I_2\,\delta\omega_1[t]\,\omega_{20}[t]\right)\right\}$$

```
w[5] = Map[Collect[#, {δω1[t], δω2[t], δω3[t], ω10[t], ω20[t], ω30[t]}] &, w[4], {2}]
```

$$\left\{\delta\omega_1'[t]\to\frac{(I_2-I_3)\,\delta\omega_3[t]\,\omega_{20}[t]}{I_1}+\frac{(I_2-I_3)\,\delta\omega_2[t]\,\omega_{30}[t]}{I_1},\right.$$
$$\delta\omega_2'[t]\to\frac{(-I_1+I_3)\,\delta\omega_3[t]\,\omega_{10}[t]}{I_2}+\frac{(-I_1+I_3)\,\delta\omega_1[t]\,\omega_{30}[t]}{I_2},$$
$$\left.\delta\omega_3'[t]\to\frac{(I_1-I_2)\,\delta\omega_2[t]\,\omega_{10}[t]}{I_3}+\frac{(I_1-I_2)\,\delta\omega_1[t]\,\omega_{20}[t]}{I_3}\right\}$$

Introduce some simplifying definitions, defined such that the κ are all positive for $I_1 > I_2 > I_3$

```
w[6] = w[5] /. {(I2 - I3) → κ1 I1, (-I1 + I3) → -κ2 I2, (I1 - I2) → κ3 I3 }
```

$$\{\delta\omega_1'[t]\to\kappa_1\,\delta\omega_3[t]\,\omega_{20}[t]+\kappa_1\,\delta\omega_2[t]\,\omega_{30}[t],$$
$$\delta\omega_2'[t]\to-\kappa_2\,\delta\omega_3[t]\,\omega_{10}[t]-\kappa_2\,\delta\omega_1[t]\,\omega_{30}[t],\ \delta\omega_3'[t]\to\kappa_3\,\delta\omega_2[t]\,\omega_{10}[t]+\kappa_3\,\delta\omega_1[t]\,\omega_{20}[t]\}$$

Case 1: Consider that the system is started rotating about the 1 direction. However, small amounts of rotation are also excited in the 2 and 3 directions

```
w11[1] = w[6] /. {ω20[t] → 0, ω30[t] → 0} /. Rule → Equal
```

$$\{\delta\omega_1'[t]==0,\ \delta\omega_2'[t]==-\kappa_2\,\delta\omega_3[t]\,\omega_{10}[t],\ \delta\omega_3'[t]==\kappa_3\,\delta\omega_2[t]\,\omega_{10}[t]\}$$

The first equation indicates that $\delta\omega_1$ is a constant, hence $\omega_{10}(t) = \omega_{10}$ is a constant

```
w11[2] = w11[1][[2 ;; 3]] /. ω10[t] → ω10
```

$\{\delta\omega_2'[t] == -\kappa_2\,\omega_{10}\,\delta\omega_3[t],\ \delta\omega_3'[t] == \kappa_3\,\omega_{10}\,\delta\omega_2[t]\}$

Solving these equation shows that $\delta\omega_1$[t] and $\delta\omega_2$[t] are oscillatory. Perturbations about axis 1 are stable.

```
w11[3] =
  DSolve[Join[w11[2], {δω2[0] == δω20, δω3[0] == δω30}], {δω2[t], δω3[t]}, t][[1]] //
   Expand;
w11[3] = Simplify[w11[3], {κ1 > 0, κ2 > 0, κ3 > 0}]
```

$$\left\{\delta\omega_2[t] \to \mathrm{Cos}\left[t\sqrt{\kappa_2\,\kappa_3}\ \omega_{10}\right]\delta\omega_{20} - \mathrm{Sin}\left[t\sqrt{\kappa_2\,\kappa_3}\ \omega_{10}\right]\delta\omega_{30}\sqrt{\frac{\kappa_2}{\kappa_3}},\right.$$
$$\left.\delta\omega_3[t] \to \mathrm{Cos}\left[t\sqrt{\kappa_2\,\kappa_3}\ \omega_{10}\right]\delta\omega_{30} + \mathrm{Sin}\left[t\sqrt{\kappa_2\,\kappa_3}\ \omega_{10}\right]\delta\omega_{20}\sqrt{\frac{\kappa_3}{\kappa_2}}\right\}$$

Case 2: Consider that the system is started rotating about the 3 direction. However, small amounts of rotation are also excited in the 1 and 2 directions

```
w12[1] = w[6] /. {ω10[t] → 0, ω20[t] → 0} /. Rule → Equal
```

$\{\delta\omega_1'[t] == \kappa_1\,\delta\omega_2[t]\,\omega_{30}[t],\ \delta\omega_2'[t] == -\kappa_2\,\delta\omega_1[t]\,\omega_{30}[t],\ \delta\omega_3'[t] == 0\}$

The third equation indicates that $\delta\omega_3$ is a constant, hence $\omega_{30}(t) = \omega_{30}$ is a constant

```
w12[2] = w12[1][[1 ;; 2]] /. ω30[t] → ω30
```

$\{\delta\omega_1'[t] == \kappa_1\,\omega_{30}\,\delta\omega_2[t],\ \delta\omega_2'[t] == -\kappa_2\,\omega_{30}\,\delta\omega_1[t]\}$

Solving these equation shows that $\delta\omega_1$[t] and $\delta\omega_2$[t] are oscillatory. Perturbations about axis 3 are stable.

```
w12[3] =
  DSolve[Join[w12[2], {δω1[0] == δω10, δω2[0] == δω20}], {δω1[t], δω2[t]}, t][[1]] /.
    C[1] → a3 /. C[2] → b3;
w12[3] = Simplify[w12[3], {κ1 > 0, κ2 > 0, κ3 > 0}] // Expand
```

$$\left\{\delta\omega_1[t] \to \mathrm{Cos}\left[t\sqrt{\kappa_1\,\kappa_2}\ \omega_{30}\right]\delta\omega_{10} + \mathrm{Sin}\left[t\sqrt{\kappa_1\,\kappa_2}\ \omega_{30}\right]\delta\omega_{20}\sqrt{\frac{\kappa_1}{\kappa_2}},\right.$$
$$\left.\delta\omega_2[t] \to \mathrm{Cos}\left[t\sqrt{\kappa_1\,\kappa_2}\ \omega_{30}\right]\delta\omega_{20} - \mathrm{Sin}\left[t\sqrt{\kappa_1\,\kappa_2}\ \omega_{30}\right]\delta\omega_{10}\sqrt{\frac{\kappa_2}{\kappa_1}}\right\}$$

Case 3: Consider that the system is started rotating about the 2 direction. However, small amounts of rotation are also excited in the 1 and 2 directions

```
w13[1] = w[6] /. {ω10[t] → 0, ω30[t] → 0} /. Rule → Equal
```

$\{\delta\omega_1'[t] == \kappa_1\,\delta\omega_3[t]\,\omega_{20}[t],\ \delta\omega_2'[t] == 0,\ \delta\omega_3'[t] == \kappa_3\,\delta\omega_1[t]\,\omega_{20}[t]\}$

The second equation indicates that $\delta\omega_2$ is a constant, hence $\omega_{20}(t) = \omega_{20}$ is a constant

```
w13[2] = w13[1][[{1, 3}]] /. ω20[t] → ω20
```

$\{\delta\omega_1'[t] == \kappa_1\,\omega_{20}\,\delta\omega_3[t],\ \delta\omega_3'[t] == \kappa_3\,\omega_{20}\,\delta\omega_1[t]\}$

Solving these equation shows that $\delta\omega_1$[t] and $\delta\omega_2$[t] have exponential terms. Perturbations about axis 2 are unstable.

```
w13[3] =
  DSolve[Join[w13[2], {δω1[0] == δω10, δω3[0] == δω30}], {δω1[t], δω3[t]}, t][[1]] ;
w13[3] = Simplify[w13[3], {κ1 > 0, κ2 > 0, κ3 > 0}] // Expand
```

$$\left\{\delta\omega_1[t] \to \frac{1}{2}e^{-t\sqrt{\kappa_1\kappa_3}\,\omega_{20}}\delta\omega_{10} + \frac{1}{2}e^{t\sqrt{\kappa_1\kappa_3}\,\omega_{20}}\delta\omega_{10} - \frac{e^{-t\sqrt{\kappa_1\kappa_3}\,\omega_{20}}\delta\omega_{30}\sqrt{\kappa_1}}{2\sqrt{\kappa_3}} + \frac{e^{t\sqrt{\kappa_1\kappa_3}\,\omega_{20}}\delta\omega_{30}\sqrt{\kappa_1}}{2\sqrt{\kappa_3}},\right.$$
$$\left.\delta\omega_3[t] \to \frac{1}{2}e^{-t\sqrt{\kappa_1\kappa_3}\,\omega_{20}}\delta\omega_{30} + \frac{1}{2}e^{t\sqrt{\kappa_1\kappa_3}\,\omega_{20}}\delta\omega_{30} - \frac{e^{-t\sqrt{\kappa_1\kappa_3}\,\omega_{20}}\delta\omega_{10}\sqrt{\kappa_3}}{2\sqrt{\kappa_1}} + \frac{e^{t\sqrt{\kappa_1\kappa_3}\,\omega_{20}}\delta\omega_{10}\sqrt{\kappa_3}}{2\sqrt{\kappa_1}}\right\}$$

2 Numerics

To supplement the analysis above, I illustrate the instability of motion about the intermediate axis by numerically solving the Euler equations

```
w2[1] = w[2] /. τa_ → 0 // Simplify
```

$$\left\{\omega_1'[t] \to \frac{(I_2 - I_3)\,\omega_2[t]\,\omega_3[t]}{I_1},\ \omega_2'[t] \to \frac{(-I_1 + I_3)\,\omega_1[t]\,\omega_3[t]}{I_2},\ \omega_3'[t] \to \frac{(I_1 - I_2)\,\omega_1[t]\,\omega_2[t]}{I_3}\right\}$$

```
w2[2] = w2[1] /. {(I2 - I3) → κ1 I1, (-I1 + I3) → -κ2 I2, (I1 - I2) → κ3 I3 }
```

$\{\omega_1'[t] \to \kappa_1\,\omega_2[t]\,\omega_3[t],\ \omega_2'[t] \to -\kappa_2\,\omega_1[t]\,\omega_3[t],\ \omega_3'[t] \to \kappa_3\,\omega_1[t]\,\omega_2[t]\}$

I don’t use subscripts inside Module

```
w2[3] = w2[2] /. {ω1 → ω1, ω2 → ω2, ω3 → ω3} /.
    {κ1 → κ1, κ2 → κ2, κ3 → κ3} /. Rule → Equal
```

$\{\omega1'[t] == \kappa1\,\omega2[t]\,\omega3[t],\ \omega2'[t] == -\kappa2\,\omega1[t]\,\omega3[t],\ \omega3'[t] == \kappa3\,\omega1[t]\,\omega2[t]\}$

ω_1 and ω_3 stay near their initial values while ω_2 quickly deviates from its initial value.

```
Module[{I1 = 3, I2 = 2, I3 = 1., ω0 = 1, ϵ = 0.1, tMax = 30., G},
 G[1] = ShowNumericalSolution[tMax,
   {ω0, ϵ, ϵ}, {I1, I2, I3}, "stable rotation about x1"];
 G[2] = ShowNumericalSolution[tMax, {ϵ, ω0, ϵ},
   {I1, I2, I3}, "unstable rotation about x2"];
 G[3] = ShowNumericalSolution[tMax, {ϵ, ϵ, ω0},
   {I1, I2, I3}, "stable rotation about x3"];
 Grid[{{G[1], G[2], G[3]}}]]
```




```
Clear[ShowNumericalSolution];
ShowNumericalSolution[tMax_, {ω10_, ω20_, ω30_}, {I1_, I2_, I3_}, lab_] :=
 Module[{scale = 2, δ = 0.1, O, e1, e2, e3, axes,
    range, κ1, κ2, κ3, eqns, initialConditions, soln, startArrow, G},
  range = 1.25 {{-1, 1}, {-1, 1}, {-1, 1}};
  {O, e1, e2, e3} = {{0, 0, 0}, {1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
  axes = {Black, Line[scale {-e1, e1}], Line[scale { -e2, e2}],
    Line[scale {- e3, e3}], Stl@Text["ω1", (scale + δ) e1],
    Stl@ Text["ω2", (scale + δ) e2], Stl@Text["ω3", (scale + δ) e3]};
  startArrow = {Red, Arrow[{O, {ω10, ω20, ω30}}]};
  G[1] = Graphics3D[{axes, startArrow},
    PlotRange → range, Boxed → False, PlotLabel → Stl@lab];

  {κ1, κ2, κ3} = {(I2 - I3) / I1, -(I3 - I1) / I2, (I1 - I2) / I3};
  eqns = {ω1'[t] == κ1 ω2[t] ω3[t], ω2'[t] == -κ2 ω1[t] ω3[t], ω3'[t] == κ3 ω1[t] ω2[t]};
  initialConditions = {ω1[0] == ω10, ω2[0] == ω20, ω3[0] == ω30};
  eqns = Join[eqns, initialConditions];
  soln = NDSolve[eqns, {ω1, ω2, ω3}, {t, 0, tMax}];
  G[2] = ParametricPlot3D[{ω1[t] , ω2[t], ω3[t]} /. soln,
    {t, 0, tMax}, PlotRange → range, Boxed → False, PlotStyle → Black];
  Show[G[1], G[2]]]
```